Gyakorlat vázlat Ujfalussy Balázs: *IDEGRENDSZERI MODELLEZÉS* c. órájához

Dinamikai rendszerek egyszerű vizsgálata – XPP

- XPP honlapja: `http://www.math.pitt.edu/~bard/xpp/xpp.html`
- Ez a program kicsit fapados, de egyszerűen kezelhető, ingyenesen letölthető és rugalmasan használható differenciálegyenletek numerikus szimulációjára. Mivel eredetileg is idegsejtek membránján lejátszódó elektromos folyamatok szimulációjára fejlesztették ki, számos tankönyvi mintafeladatot könnyen elérhető hozzá.
- Ismerendő fogalmak: *Differenciál hányados, differenciál egyenlet, fix pontok (stabil, instabil), határciklus (stabil, instabil).*
- Mi ezeket mind most nem tárgyaljuk. A célunk csupán az, hogy a hallgatók ízelítőt kapjanak a dinamikus rendszerek vizsgálatából és egy kicsit maguk is kipróbálják azokat. A Hodgkin-Huxley modellt fogjuk vizsgálni.

A Hodgkin-Huxley modell

A Hodgkin-Huxley egyenletek:

$$V' = (I + g_{Na} m^3 h (V_{Na} - V_m) + g_K n^4 (V_K - V_m) + g_L (V_L - V_m))/C \tag{1}$$

$$m' = \alpha_m(V_m)(1-m) - \beta_m(V_m) m \tag{2}$$

$$h' = \alpha_h(V_m)(1-h) - \beta_h(V_m) h \tag{3}$$

$$n' = \alpha_n(V_m)(1-n) - \beta_n(V_m) n, \tag{4}$$

ahol g_{Na}, g_K, g_L és V_L a membrán áteresztőképessége a nátrium, kálium és egyéb (főként Cl^-) ionokra, és azok egyensúlyi potenciálja. m, h és n az ioncsatornák kapuváltozói, α és β pedig a kapu – membránpotenciáltól függő – nyitási és zárási sebessége.

File: `/home/student/Programs/Xpp/HH.ode`
Help: `/home/student/Programs/Xpp/help/xpp_doc.pdf`

```
>---------------------- Nyissz -----------------------<
# Hodgkin-huxley egyenletek
v'=(I - gna*h*(v-vna)*m^3-gk*(v-vk)*n^4-gl*(v-vl))/c
m'= am(v)*(1-m)-bm(v)*m
h'=ah(v)*(1-h)-bh(v)*h
n'=an(v)*(1-n)-bn(v)*n
# units: mV     mV     mV    mS/cm2            uF/cm2, uA/cm2
par vna=50,vk=-77,vl=-54.4,gna=120,gk=36,gl=.3,c=1,i=0
am(v) =  .1*(v+40)/(1-exp(-(v+40)/10))
bm(v) =  4*exp(-(v+65)/18)
ah(v) =  .07*exp(-(v+65)/20)
bh(v) =  1/(1+exp(-(v+35)/10))
an(v) =  .01*(v+55)/(1-exp(-(v+55)/10))
bn(v) =  .125*exp(-(v+65)/80)
  init v=-65,m=.052,h=.596,n=.317
aux gka=gk*n^4
aux gnat=gna*h*m^3
aux Aram=i
@ total=100
@ xplot=t,yplot=v
@ xlo=0,ylo=-100,xhi=100,yhi=40
@ bound=1000000,MAXSTOR=2000002
done
<---------------------- Nyissz ----------------------->
```

Számítógép használata

- Bejelentkezés: felhasználó: `student`; jelszó: `Alma1`
- terminal: `Alt + Ctr + t`
- lepj be az XPP kódot tartalmazó könyvtárba `cd /home/student/Programs/Xpp/`

Program használata

- File futtatása: `xppaut HH.ode`
- Numerikus integrálás: `ig`
- Integrálás folytatása: `il`
- Kék [Param] gombocska – parameterek állítása. `enter`: tovább, `tab`: elfogad
- `v2`: a tengelyek beállítása
- `ga`: görbe hozzáadása
- `wf`: az ablak „fittelése”
- `e`: az aktuális ábra törlése

Feladatok I.

Az ingerküszöb megkeresése. Növeljük a serkentőáramot ($I = 0 \rightarrow 12\ \mu S/cm^2$, egyesével), indítsuk el a szimulációt (`ig`) és figyeljük a sejt reakcióját! Legegyszerűbben így változtathatjuk a kívánt paramétert: A program-ablak alján van 3 fülecske `Par/Var?` felirattal. Ha ide beírjuk a kívánt paramétert, akkor a toszattyúval állíthatjuk az értékét. Mikor kezd el tüzelni a sejt? És mekkora áram után kezd folyamatosan tüzelni? Akkor mekkora a küszöb?

Most kezdjük előröl, de úgy, hogy a paraméter értékének beállítása után a szimulációt az `il` paranccsal indítjuk. Így az előző értéket megtartjuk. Keressük meg a küszöböt, majd kezdjük fokozatosan csökkenteni az áramot. Mit tapasztalunk?

Feladatok II.

Nézzük meg most a dinamikai rendszer különböző változóit (V, m, h, n) az idő függvényében, majd az egyes változókat egymás függvényében ($V - m, V - n, n - h, \dots$) is egy akciós potenciál sorozat alatt. Idézzük fel, hogy mi az egyes kapuk szerepe az akciós potenciál kialakulásakor!

A HH-rendszer meglehetősen bonyolult, 4 dimenziós dinamikai rendszer. Szükség van-e mind a 4 dimenzióra? Nem lehetne-e valahogyan egyszerűsíteni a rendszert? Figyeld meg például az n-kapu nyitottságát a h-kapu függvényében!